
HP-33S Calculator Program Closure 8

Crandall’s Adjustment for a Closed Traverse

Programmer: Dr. Bill Hazelton

Date: June, 2007.

Line Instruction Display User Instructions
K0001 LBL  K
K0002 CL Σ
K0003 SF  10 (Use  FLAGS  SF  .0)
K0004 ENTER  FIX  SIDES (Use  EQN  RCL E, RCL N, RCL T, etc.)
K0005 PSE
F0001 LBL  F
F0002 ENTER  F  AZ (Use  EQN  RCL E, RCL N, RCL T, etc.)
F0003 PSE
F0004 INPUT  Q
F0005 RCL  Q
F0006 x < 0?
F0007 GTO  V
F0008 RCL  Q
F0009 →HR
F0010 STO  Q
F0011 ENTER  F  DIST (Use  EQN  RCL E, RCL N, RCL T, etc.)
F0012 PSE
F0013 INPUT  D
F0014 RCL  Q
F0015 RCL  D
F0016 θ,r → y,x
F0017 Σ+
F0018 GTO  F
V0001 LBL  V
V0002 31
V0003 STO  i i is on decimal point key.
V0004 0
V0005 STO (i) (i) is on ENTER key.
V0006 32
V0007 STO  i
V0008 0
V0009 STO (i)
V0010 33
V0011 STO  i
V0012 0
V0013 STO (i)
V0014 ENTER  VAR  SIDES (Use  EQN  RCL E, RCL N, RCL T, etc.)
V0015 PSE
U0001 LBL  U



HP-33S Calculator Program Closure   8
Crandall’s Adjustment for a Closed Traverse

- 2 -
 -

Line Instruction Display User Instructions
U0002 ENTER  V  AZ (Use  EQN  RCL E, RCL N, RCL T, etc.)
U0003 PSE
U0004 INPUT  Q
U0005 RCL  Q
U0006 x < 0?
U0007 GTO  J
U0008 RCL  Q
U0009 →HR
U0010 STO  Q
U0011 ENTER  V  DIST (Use  EQN  RCL E, RCL N, RCL T, etc.)
U0012 PSE
U0013 INPUT  D
U0014 RCL  Q
U0015 RCL  D
U0016 θ,r → y,x
U0017 Σ+
U0018 GTO  U
J0001 LBL  J
J0002 Σx2 (Enter using  SUMS  Σx2)
J0003 Σy2 (Enter using  SUMS  Σy2)
J0004 ×
J0005 Σxy (Enter using  SUMS  Σxy)
J0006 x2

J0007 –
J0008 STO  D
J0009 Σy (Enter using  SUMS  Σy)
J0010 Σxy (Enter using  SUMS  Σxy)
J0011 ×
J0012 Σx (Enter using  SUMS  Σx)
J0013 Σy2 (Enter using  SUMS  Σy2)
J0014 ×
J0015 –
J0016 RCL÷  D
J0017 STO  A
J0018 Σx (Enter using  SUMS  Σx)
J0019 Σxy (Enter using  SUMS  Σxy)
J0020 ×
J0021 Σy (Enter using  SUMS  Σy)
J0022 Σx2 (Enter using  SUMS  Σx2)
J0023 ×
J0024 –
J0025 RCL÷  D
J0026 STO  B
J0027 0
J0028 STO  V
J0029 ENTER  VAR  SIDES (Use  EQN  RCL E, RCL N, RCL T, etc.)

Line Instruction Display User Instructions
J0030 PSE
L0001 LBL  L
L0002 ENTER  V  AZ (Use  EQN  RCL E, RCL N, RCL T, etc.)
L0003 PSE
L0004 INPUT  Q
L0005 RCL  Q
L0006 x < 0?
L0007 GTO  M
L0008 RCL  Q
L0009 →HR
L0010 STO  Q
L0011 ENTER  V  DIST (Use  EQN  RCL E, RCL N, RCL T, etc.)
L0012 PSE
L0013 INPUT  D
L0014 RCL  Q
L0015 RCL  D
L0016 θ,r → y,x
L0017 RCL×  D
L0018 RCL×  A
L0019 x < > y
L0020 RCL×  D
L0021 RCL×  B
L0022 +
L0023 STO  C
L0024 STO+  V
L0025 DIST  CORRN (Use  EQN  RCL D, RCL I, RCL S, etc.)
L0026 PSE
L0027 VIEW  C
L0028 RCL  C
L0029 STO+  D
L0030 CORRD  DIST (Use  EQN  RCL C, RCL O, RCL R, etc.)
L0031 PSE
L0032 VIEW  D
L0033 GTO  L
M0001 LBL  M
M0002 SUM  OF  CORRNS (Use  EQN  RCL S, RCL U, etc.)
M0003 PSE
M0004 VIEW  V
M0005 CF  10 (Use  FLAGS  CF  .0)
M0006 RTN

Notes

(1) This program uses a lot of labels (7), because of the need to loop through side
entry and to work through several choices. Consequently, it should be entered
only if it is really needed.

(2) The program can work with traverses where all the sides are to be adjusted,
where some of the sides are considered ‘fixed’ (not to be adjusted), and traverses
between known points with co-ordinates. The last need the azimuth and distance
between the fixed points derived from the co-ordinates, and treated as a fixed
side. (Use the Utility 6 program or Utility 4 key sequence, if needed.)

(3) The sequence of operations is to enter the fixed sides (those not to be adjusted),
then the variable sides (those to be adjusted), and then the variable sides again.
The reason for entering the variable sides twice is that the calculator does not
have sufficient memory to store the sides for processing a second time, so they
must be re-entered.

(4) Provided that all the fixed sides are entered first, then all the variable sides, then
all the variable sides again, it does not matter in which order the sides within
each group are entered.

(5) Azimuths are entered in HP notation, i.e., DDD.MMSS

(6) The misclosure components of the unadjusted traverse in X (or E) and Y (or N)
can be displayed by recalling Σy and Σx using the SUMS menu. (Note these are
back-to-front.) Reversing their order using x<>y sets them up for conversion to
polar. To check the traverse closes after adjustment, enter the sides through one
of the other traverse programs, e.g., Closure 1 or Closure 2.

(7) This program was inspired by the program developed by Neil H. Bradbury, LS,
in January, 1977, for the HP-25 calculator.

(8) The program assumes that the error in the length of each side is proportional to
the length of the side, which is the situation with EDM. It will produce somewhat
different results to a Crandall’s adjustment based on the error increasing as the
square root of the distance, which is the more traditional method based on error
propagation using a tape.

(9) To indicate the end of the sides of the traverse to be entered at any point, enter a
negative value for the azimuth, –1 is easiest.

Theory

Crandall’s Adjustment was developed by Charles Crandall, and published in 1914. It was the first
serious rival to Bowditch’s Method of adjustment. The fundamental concepts were that angle
measurement was far better than distance measurement, and that the small random errors in the
angles of a traverse could be properly adjusted by simple distribution of the angular misclosure
among the traverse’s angles. This meant that all the remaining misclosure in the traverse, i.e., all
the random errors, could be accounted for by the distances alone, so these were all that was
adjusted by Crandall’s Adjustment.

Crandall used least squares adjustment by condition equations for the adjustment. By avoiding
having to adjust the angles as part of the adjustment proper, the number of conditions was
reduced to two, which greatly simplified the calculations.

Angle Adjustment

The angular misclosure of the traverse can be adjusted by any reasonable means. The misclosure
can be distributed evenly among all the angles, or larger amounts may be apportioned to angles
that can be assumed to be less reliable, perhaps because of short lines or poor lies of sight.
However the angles are adjusted, they should be adjusted to bring the misclosure to zero, and then
the azimuths of each line calculated.

It is a good idea to run the traverse through one of the closure programs to make sure that there
are no gross or systematic errors in the traverse, prior to adjustment. This is because the
adjustment process assumes that only random errors are present in the data. If this assumption is
not correct, the adjustment will produce erroneous results.

Distance Adjustment

With the angular misclosure adjusted to zero, there are now only two conditions that must be met
to have a consistent traverse. (The third condition, that the angular misclosure is zero, has been
met by adjusting the angle ahead of this point.) Solution of a 2 × 2 matrix is straightforward, so a
full least squares adjustment by condition equations can be performed. This adjustment changes
only the lengths of the sides of the traverse. The angles are already adjusted and so held fixed.

If the length of each line in the traverse is Mi, while θi is its azimuth, then di will be the adjusted
length of each line in the traverse, and vi will be the correction to the measured length of each line
to obtain the adjusted length:

di  =  Mi  +  vi [1]

The conditions to be met can be expressed as:

d1 cos θ1  +  d2 cos θ2  +  d3 cos θ3  +  …  =  0 [Sum of latitudes equals zero]

d1 sin θ1  +  d2 sin θ2  +  d3 sin θ3  +  …  =  0 [Sum of departures equals zero]

Substituting Equation [1] into the above equations gives:

(M1 + v1) cos θ1  +  (M2 + v2) cos θ2  +  (M3 + v3) cos θ3  +  …  =  0

(M1 + v1) sin θ1  +  (M2 + v2) sin θ2  +  (M3 + v3) sin θ3  +  …  =  0

Re-arranging and gathering terms produces:

v1 cos θ1  +  v2 cos θ2  +  v3 cos θ3  +  …  +  q1  =  0

v1 sin θ1  +  v2 sin θ2  +  v3 sin θ3  +  …  +  q2  =  0 [2]

where q1  = M1 cos θ1  +  M2 cos θ2  +  M3 cos θ3  +  …  =  misclosure in latitude

q2  = M1 sin θ1  +  M2 sin θ2  +  M3 sin θ3  +  …  =  misclosure in departure

So q1 and q2 are obtained directly from the closure check of the traverse, done after the angles are
adjusted and azimuths computed. Representing the departure of each line by Di, and the latitude
of each line by Li, Equations [2] can be rewritten as:

v1 cos 
L1
d1

  +  v2 cos 
L2
d2

  +  v3 cos 
L3
d3

  +  …  +  q1  =  0

v1 sin 
D1
d1

  +  v2 sin 
D2

d2
  +  v3 sin 

D3

d3
  +  …  +  q2  =  0 [3]

The theory of least squares now allows the development of what are termed the normal equations
(named after the normal distribution, which describes the distribution characteristics of the

random errors). The details can be found elsewhere, but basically they are a combination of all
the measured data, the precision values of the measurements, and the models that describe how
they all fit together, i.e., the model of the conditions to be met (e.g., Equations [2] and [3]).

There are two cases that Crandall considered. The first is the situation where the error in the
length of the sides increases in proportion to the square root of the length of the line, i.e., d ,
which is a good assumption for taped distances. The second case is where the error in distance
increases in proportion to the length of the line, which is more the case with short distances
measured with rods, but also the case with EDM. This program is based on the second case, as it
is more probable that EDM traverses will be used with this program.

If the error in the lengths of the sides of the traverse increases in proportion to the length of each
side, then the normal equations for the adjustment are as follows:

A Li
2

i
∑ + B LiDi

i
∑ + q1 = 0

A LiDi
i
∑ + B Di

2

i
∑ + q2 = 0 [4]

Solving for A and B, which are called the correlatives:

A  = 

� 

q2 LiDi
i
∑ − q1 Di

2

i
∑

Di
2 Li

2

i
∑ − LiDi

i
∑
⎛ 

⎝ 
⎜ 

⎞ 

⎠ 
⎟ 
2

i
∑

B  =  

� 

q1 LiDi
i
∑ − q2 Li

2

i
∑

Di
2 Li

2

i
∑ − LiDi

i
∑
⎛ 

⎝ 
⎜ 

⎞ 

⎠ 
⎟ 
2

i
∑

[5]

the corrections to the line lengths can be calculated using:

vi  =  Li  di  A  +  Di  di  B [6]

As an aside, the corrections should all sum to zero if the traverse has no fixed sides, but that can
be affected by rounding during the calculations, especially if the sums of the various components
are large, so the result may be not quite zero. If the traverse has fixed sides, the sum of the
corrections will be significantly different from zero depending upon the length of the fixed side.

Including Fixed Sides and Points

In the event that the traverse includes more than one ‘fixed’ point (usually the starting point in
calculations), Crandall’s Method can still be used. A ‘fixed’ point means a point that has known
co-ordinates, usually from a more precise prior survey, and will not be subject to adjustment in
this program. This situation arises when a traverse connects to more than one known point.

If there are two fixed points, and these form the end points of one line of the traverse, then the
traverse calculations are done in the normal way, except that the Li, Di, and LiDi values for the
fixed line are not calculated and not included in the summations to solve the correlatives. This is
the case in this program.

If there are more than two fixed points, or the fixed points are not adjacent, the traverse must be
split into sections between fixed points, and the sections adjusted independently. This is not the
ideal solution, as a full least squares adjustment should be used in this situation.

If the traverse runs between two known points, then the line (vector) between the two known
points (computed from the co-ordinates of the two points) can be considered another side in the
traverse. Computing around the traverse, the misclosure will reflect how well the traverse agrees
with this computed vector. To adjust the traverse, don’t include the Li, Di, and LiDi values for this
line in the summations, and all the corrections will occur in the measured traverse lines.

For this program, any fixed lines should be worked out ahead of time, including the line between
known end-points of the traverse (if applicable). The first part of the program allows the user to
enter these sides, and they do not add to the data that sets up the adjustment. The second part of
the program allows the user to enter the sides that do add to the adjustment, and these are entered
again for the actual adjustment process.

Sample Computations and Running the Program

1. All sides to be adjusted

The traverse to be adjusted is as follows:

Azimuth Distance

0° 12' 156.41

90° 00' 211.65

165° 49' 173.82

250° 55' 176.60

308° 30' 112.26

The initial misclosure is: DE = 0.036

DN = 0.033

Misclosure Length = 0.049

Misclosure Azimuth =    47° 24' 15"

Running the Program

Press  XEQ  K to start the program.

The program shows ENTER FIX SIDE to indicate that the fixed sides should be entered first.

The program then prompts ENTER F AZ, to indicate that the azimuth of a fixed side should be
entered in DDD.MMSS (HP) notation. The prompt is to enter Q?.

As there are no fixed sides in this case, key in  1  then press the +/–  key, and press R/S.

The calculator shows ENTER VAR SIDE to indicate that the sides to be varied (adjusted) should
be entered now.

The calculator then prompts  ENTER V AZ to indicate that the azimuth of a variable sides should
be entered in DDD.MMSS (HP) notation. The prompts to enter is Q?.

Key in the azimuth of the first side,  0.12, then  press R/S.

The calculator prompts with ENTER V DIST, to indicate that the length of that variable side
should be entered. The prompt is to enter D?.

Key in the distance  156.41, then press R/S.

The calculator prompts with ENTER V AZ, then the Q? prompt.

Key in the next azimuth, 90.0, and press R/S.

The calculator prompts with ENTER V DIST, then the D? prompt.

Key in the distance,  211.65, and press R/S.

The calculator prompts with ENTER V AZ, then the Q? prompt.

Key in the next azimuth, 165.49, and press R/S.

The calculator prompts with ENTER V DIST, then the D? prompt.

Key in the distance,  173.83, and press R/S.

The calculator prompts with ENTER V AZ, then the Q? prompt.

Key in the next azimuth, 250.55, and press R/S.

The calculator prompts with ENTER V DIST, then the D? prompt.

Key in the distance,  176.6, and press R/S.

The calculator prompts with ENTER V AZ, then the Q? prompt.

Key in the next azimuth, 308.3, and press R/S.

The calculator prompts with ENTER V DIST, then the D? prompt.

Key in the distance,  112.26, and press R/S.

The calculator prompts with ENTER V AZ, then the Q? prompt.

As this is the last variable side, key in a negative azimuth, –1, and press R/S.

The calculator prompts with ENTER VAR SIDE, and prepares to receive the variable sides again
for the adjustment calculation.

The calculator prompts with ENTER V AZ, then the Q? prompt.

Key in the azimuth of the first side, 0.12, then press R/S.

The calculator prompts with ENTER V DIST, then the D? prompt.

Key in the length of the first side, 156.41, then press R/S.

The calculator shows DIST CORRN, indicating the correction to the length of that side, then
shows the correction, using the C= prompt. In this case, it is –0.0139. Press R/S.

The calculator shows CORRD DIST, indicating the corrected (adjusted) distance of the side, then
shows the corrected distance, using the D= prompt. In this case, it is 156.3961. Press R/S.

The calculator prompts with ENTER V AZ, then the Q? prompt.

Key in the azimuth of the next side, 90, then press R/S.

The calculator prompts with ENTER V DIST, then the D? prompt.

Key in the length of the side, 211.65, then press R/S.

The calculator shows DIST CORRN, indicating the correction to the length of that side, then
shows the correction, using the C= prompt. In this case, it is –0.0207. Press R/S.

The calculator shows CORRD DIST, indicating the corrected (adjusted) distance of the side, then
shows the corrected distance, using the D= prompt. In this case, it is 211.6293. Press R/S.

The calculator prompts with ENTER V AZ, then the Q? prompt.

Key in the azimuth of the next side, 165.49, then press R/S.

The calculator prompts with ENTER V DIST, then the D? prompt.

Key in the length of the side, 173.82, then press R/S.

The calculator shows DIST CORRN, indicating the correction to the length of that side, then
shows the correction, using the C= prompt. In this case, it is 0.0132. Press R/S.

The calculator shows CORRD DIST, indicating the corrected (adjusted) distance of the side, then
shows the corrected distance, using the D= prompt. In this case, it is 173.8332. Press R/S.

The calculator prompts with ENTER V AZ, then the Q? prompt.

Key in the azimuth of the next side, 250.55, then press R/S.

The calculator prompts with ENTER V DIST, then the D? prompt.

Key in the length of the side, 176.6, then press R/S.

The calculator shows DIST CORRN, indicating the correction to the length of that side, then
shows the correction, using the C= prompt. In this case, it is 0.0194. Press R/S.

The calculator shows CORRD DIST, indicating the corrected (adjusted) distance of the side, then
shows the corrected distance, using the D= prompt. In this case, it is 176.6194. Press R/S.

The calculator prompts with ENTER V AZ, then the Q? prompt.

Key in the azimuth of the last side, 308.3, then press R/S.

The calculator prompts with ENTER V DIST, then the D? prompt.

Key in the length of the last side, 112.26, then press R/S.

The calculator shows DIST CORRN, indicating the correction to the length of that side, then
shows the correction, using the C= prompt. In this case, it is 0.0001. Press R/S.

The calculator shows CORRD DIST, indicating the corrected (adjusted) distance of the side, then
shows the corrected distance, using the D= prompt. In this case, it is 112.2601. Press R/S.

The calculator prompts with ENTER V AZ, then the Q? prompt.

As this is the last side, key in a negative azimuth, –1, then press R/S.

The calculator shows  SUM OF CORRNS, then shows the sum of the corrections to the distances,
using the V= prompt. In this case, the sum is –0.0019, which is reasonably close to zero.

Pressing R/S again, resets the flags and ends the program.

Note that the sides in each group may be entered in any order. When they variable sides are
entered a second time to calculate the corrections to their lengths, sides may be entered more than
once, e.g., to double-check the results, but this will lead to errors in the sum of the corrections
value displayed at the end of the program.

Summarizing the results, the traverse is now adjusted to be:

Azimuth Adjusted Distance

0° 12' 156.3961

90° 00' 211.6293

165° 49' 173.8332

250° 55' 176.6194

308° 30' 112.2601

The final misclosure is: DE = –0.000 015

DN = 0.000 040

Misclosure Length = 0.000 042

Misclosure Azimuth =    339° 35'

The misclosure is now smaller than the finest distance measurement, so all that remains is
rounding error in the calculations.

2. One fixed side, a traverse between known points

For this traverse, one side is considered fixed, because the traverse is between two fixed points.
The azimuth and distance between the two points are deduced from co-ordinates (using a
different program), and are 59° 04' 33" for 1995.78. The angular misclosure is adjusted out of the
traverse angles, and the azimuths of the sides calculated, on the same azimuth datum as the fixed
side, i.e., the same co-ordinate system.

The program is started and the fixed side entered: 59.0433 and 1995.78.

As this is the only fixed side, the next azimuth is entered as –1, and the calculator moves to
collect the data for the variable sides. These are as follows:

Azimuth Distance

349° 47' 19" 429.20

109° 22' 12" 476.00

202° 58' 13" 1362.00

271° 07' 59" 210.10

259° 42' 45" 164.00

295° 34' 53" 319.80

237° 45' 39" 499.95

283° 57' 32" 482.10

The initial misclosure is: DE = 2.910

DN = –1.290

Misclosure Length = 3.183

Misclosure Azimuth =    113° 54' 23"

These are entered into the calculator as prompted. Once they have all been entered, an azimuth of
–1 is entered, then the sides are entered again, for the adjustment results to be calculated. The
adjusted sides are:

Azimuth Distance Correction Distance

349° 47' 19" 0.4683 429.668

109° 22' 12" –0.9666 475.033

202° 58' 13" –0.4317 1361.568

271° 07' 59" 0.1719 210.272

259° 42' 45" 0.0930 164.093

295° 34' 53" 0.4393 320.239

237° 45' 39" 0.5639 500.514

283° 57' 32" 0.9761 483.076

Entering –1 as the next azimuth, the calculator displays 1.314 as the sum of the corrections.

The final misclosure is: DE = 0.000 081

DN = –0.000 115

Misclosure Length = 0.000 141

Misclosure Azimuth =    144° 53'

The misclosure is now smaller than the finest level of distance measurement.

Comments

An important point to note about Crandall’s Adjustment is that because all the corrections are
placed in the distances, any non-random errors (e.g., gross errors) in the angles will produce very
large corrections in the distances. This is an indication that there may be gross errors in the
angles, and this should be checked.

In the second example, the original lengths were in many cases almost approximate in their
apparent precision, while the misclosure was quite large, being about 1: 1,900. This led to the
large corrections in the distances seen in the results.

Note also that in the second example, with the fixed side, the sum of the corrections was 1.314,
well away from the zero. This is characteristic of adjusting traverses with fixed sides.

Storage Registers Used

A Correlative A

B Correlative B

C Correction to side length

D Current entered side length, intermediate result in adjustment calculation, adjusted length

Q Current entered side azimuth

V Sum of corrections to the lengths

i Used for indirect addressing of the statistical registers

Statistical Registers: Σx  =  Sum of the latitudes

Σy  =  Sum of the departures

Σx2  =  Sum of the squares of the latitudes of the sides to be adjusted

Σy2  =  Sum of the squares of the departures of the sides to be adjusted

Σxy  =  Sum of the products of the latitudes and departures of the sides to
be adjusted

n    =  Number of sides entered from start

Labels Used

Label K Length  =   30 Checksum  =  2762

Label F Length  =   76 Checksum  =  10B7

Label V Length  = 132 Checksum  =  F531

Label U Length  =   76 Checksum  =  DBF0

Label J Length  = 117 Checksum  =  22E0

Label L Length  =  141 Checksum  =  6938

Label M Length  =    31 Checksum  =  0B96

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computations to check proper operation after entry.

